/*
Elise Tak	
Title: Gender Inequality in Product Markets


HYPOTHESES


Stated-Hyp1: In a male-typed product market, a product made by a woman will be
evaluated at a discount compared to the same product made by a man.

	Test-Hyp1: In beer market, product made by woman vs. man will be less likely to be purchased
	Test-Hyp2: In beer market, product made by woman vs. man will fetch lower price (wtp)
	Test-Hyp3: In beer market, product made by woman vs. man will be expected to taste worse
	Test-Hyp4: In beer market, product made by woman vs. man will be expected to have worse quality

Stated-Hyp2: A product made by a man in a female-typed market will receive a 
smaller discount than the same product made by a woman in a male-typed market.

	/* hypothesis doesn't make sense (i.e. how can cupcake be the "same product 
	in a male-typed market"). I'm assuming that this hypothesis just means that
	women making beer will get discounted more than men making cupcakes.
	*/

	Test-Hyp5: Interaction of gender x product will be significant for likelihood of purcase
	Test-Hyp6: Interaction of gender x product will be significant for wtp
	Test-Hyp7: Interaction of gender x product will be significant for expected taste
	Test-Hyp8: Interaction of gender x product will be significant for quality

	Stated-Hyp3: In a male-typed market, the evaluations of products made by women 
will experience greater gains from external status conferral compared with 
the gains experienced by men.

	Test-Hyp9: In beer market, interaction of gender x award will be higher for women in terms of likelihood of purchase. 
	Test-Hyp10: In beer market, interaction of gender x award will be higher for women in terms of wtp.
	Test-Hyp11: In beer market, interaction of gender x award will be higher for women in terms of expected taste.
	Test-Hyp12: In beer market, interaction of gender x award will be higher for women in terms of expected quality.

Stated-Hyp4: In a female-typed market, status conferral will have a less 
gender differentiated effect on evaluations of products than in a male-typed 
market.

	Test-Hyp13: Interaction of award xproduct x gender will be significant for product quality
	Test-Hyp14: Interaction of award xproduct x gender will be significant for wtp
	Test-Hyp15: Interaction of award xproduct x gender will be significant for expected taste
	Test-Hyp16: Interaction of award xproduct x gender will be significant for expected quality


*/

clear all
use "TESS004_Tak_27July17.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* is product maker male?
	gen pmaker_male = (pmaker == 1) if pmaker < . // 1 = David, 2 = Sarah

* is product beer or cupcake?
	gen beer = (pitem == 1) if pitem < . // 1 = beer, 2 = cupcake
	gen cupcake = (pitem == 2) if pitem < . 
	
* did product win an award?
	recode paward (1=1) (2=0) (*=.), gen(award)
	
	
* OUTCOME MEASURES

	* note: identical questions asked for cupcake and beer stored separately in
	* dataset.  combining them into one measure below.

* 1. how likely are you to purchase item?
	clonevar purchase = Q1
	replace purchase = Q10 if Q10 < .
	ta purchase

	* recode missing 
	replace purchase = . if purchase == 98
	ta purchase

* 2. how much are you willing to pay?
	clonevar wtp = Q2
	replace wtp = Q11 if Q11 < .
	ta wtp

	* recode missing
	replace wtp = . if wtp == 998
	ta wtp

	* remove as outliers if >= 25
	replace wtp = . if wtp >= 25
	ta wtp
	
* 3. overall, how do you expect this ___ to taste
	clonevar taste = Q4
	replace taste = Q13 if Q13 < .
	ta taste

	* recode missing
	replace taste = . if taste == 98
	ta taste
	
	lab def taste 1 "Much worse than most" 7 "Much better than most"
	lab val taste taste

* 4. what do you expect the overall quality of this ____ to be?

	clonevar quality = Q5
	replace quality = Q14 if Q14 < .
	ta quality
	
	* recode missing
	replace quality = . if quality == 98
	
********************************************************************************

* ANALYSIS

	* Test-Hyp1: In beer market, product made by woman vs. man will be less likely to be purchased
	reg purchase i.pmaker_male if beer==1
		// reject
	tess 1.pmaker_male +, init(Tak862) bonf(4)	
		
	* Test-Hyp2: In beer market, product made by woman vs. man will fetch lower price (wtp)
	reg wtp i.pmaker_male if beer==1	
		// reject
	tess 1.pmaker_male +,  bonf(4)	
	
	* Test-Hyp3: In beer market, product made by woman vs. man will be expected to taste worse
	reg taste i.pmaker_male if beer==1	
		// reject
	tess 1.pmaker_male +, bonf(4)	
	
	* Test-Hyp4: In beer market, product made by woman vs. man will be expected to have worse quality
	reg quality i.pmaker_male if beer==1
		// reject
	tess 1.pmaker_male +, bonf(4)	
	

	*Test-Hyp5: Interaction of gender x product will be significant for likelihood of purcase
	reg purchase i.pmaker_male##i.beer
			// reject
	tess 1.pmaker_male#1.beer, bonf(4)		
	
	*Test-Hyp6: Interaction of gender x product will be significant for wtp
	reg wtp i.pmaker_male##i.beer
		// reject	
	tess 1.pmaker_male#1.beer, bonf(4)				
		
	*Test-Hyp7: Interaction of gender x product will be significant for expected taste
	reg taste i.pmaker_male##i.beer
		// reject
	tess 1.pmaker_male#1.beer, bonf(4)				
		
	*Test-Hyp8: Interaction of gender x product will be significant for quality
	reg quality i.pmaker_male##i.beer
		// reject
	tess 1.pmaker_male#1.beer, bonf(4)		
	
	
		
	*Test-Hyp9: In beer market, interaction of gender x award will be higher for women in terms of likelihood of purchase. 
	reg purchase i.pmaker_male##i.award if beer==1
		// reject
	tess 1.pmaker_male#1.award -, bonf(4)		
	
	*Test-Hyp10: In beer market, interaction of gender x award will be higher for women in terms of wtp.
	reg wtp i.pmaker_male##i.award if beer==1
		// reject
	tess 1.pmaker_male#1.award -, bonf(4)		
	
	*Test-Hyp11: In beer market, interaction of gender x award will be higher for women in terms of expected taste.
	reg taste i.pmaker_male##i.award if beer==1
		// reject
	tess 1.pmaker_male#1.award -, bonf(4)		
	
	*Test-Hyp12: In beer market, interaction of gender x award will be higher for women in terms of expected quality.
	reg quality i.pmaker_male##i.award if beer==1
		// reject
	tess 1.pmaker_male#1.award -, bonf(4)		
	
	
	*Test-Hyp13: Interaction of award xproduct x gender will be significant for product quality
	regress purchase i.pmaker_male##i.award##i.beer
		// reject
	tess 1.pmaker_male#1.beer#1.award, bonf(4)		
	
	*Test-Hyp14: Interaction of award xproduct x gender will be significant for wtp
	regress wtp i.pmaker_male##i.award##i.beer
		// reject
	tess 1.pmaker_male#1.beer#1.award, bonf(4)	
	
	*Test-Hyp15: Interaction of award xproduct x gender will be significant for expected taste
	regress taste i.pmaker_male##i.award##i.beer
		// reject
	tess 1.pmaker_male#1.beer#1.award, bonf(4)	
	
	*Test-Hyp16: Interaction of award xproduct x gender will be significant for expected quality
	regress quality i.pmaker_male##i.award##i.beer
		// reject	
	tess 1.pmaker_male#1.beer#1.award, bonf(4)	
